
解集合プログラミングを用いたハミルトン閉路問題の
解法に関する考察

平手 貴大　宋 剛秀　田村 直之　番原 睦則
本稿では，解集合プログラミング (ASP) を用いたハミルトン閉路問題，最短ハミルトン閉路問題，コスト制約付き
ハミルトン閉路問題の解法について述べる．ハミルトン閉路問題を解く ASP 符号化として，undirected，directed，
acyclicity の 3 つを考案した．undirected は，ハミルトン閉路問題を次数制約と連結制約で簡潔に表現した符号化で
ある．directed は，与えられた無向グラフの各辺に対して，2 つの弧を対応させることで有向グラフ化して解く符号
化である．acyclicity は，directed符号化をベースに，連結制約に代わる部分閉路禁止制約を，ASP の組込み非閉路
制約で表現した符号化である．最短ハミルトン閉路問題とコスト制約付きハミルトン閉路問題については，考案した
3 つの符号化に目的関数とコスト制約をそれぞれ追加することで自然に拡張できる．考案した符号化の有効性を評価
するために，Flinders Hamiltonian Cycle Project で公開されているハミルトン閉路問題のインスタンス (全 1001

問) を用いて実験を行った．その結果，directed 符号化が，他の符号化と比較して，より多くの問題を高速に解くこ
とができ，その優位性が確認できた．

1 はじめに
ハミルトン閉路問題 (Hamiltonian Cycle Problem;

HCP) は，与えられたグラフの全頂点をちょうど一
度ずつ通る閉路が存在するかどうかを判定する問題
である [6]．ハミルトン閉路問題は代表的な NP 完全
問題である．ハミルトン路問題 (Hamiltonian Path

Problem; HPP) は，ハミルトン閉路問題から始点と
終点が一致するという閉路の条件を取り除いたもの
である．これらの問題は，重要な工学的応用が数多く
存在するため，古くから盛んに研究されている．例え
ば，数理最適化の分野で有名な巡回セールスマン問
題は，グラフの辺に距離が付随しているとき，最短距
離のハミルトン閉路を求める最短ハミルトン閉路問
題と考えることができる．また，ごく最近では，距離

の総和が所与の閾値以下 (または以上)であることを
制約条件として付加したコスト制約付きハミルトン
閉路問題の解を高速に全列挙するアルゴリズムも提
案されている [7]．本研究では，これらの (無向グラフ
上の)ハミルトン閉路問題およびその関連問題を対象
とする．
解集合プログラミング (Answer Set Programming;

ASP [1] [4] [8] [11]) は，論理プログラミングから派生し
た宣言的問題解法のためのプログラミングパラダイ
ムである．ASP言語は，一階論理に基づく知識表現
言語の一種である．論理プログラムは ASP のルール
の有限集合である．ASP システムは論理プログラム
から安定モデル意味論に基づく解集合を計算するシ
ステムである．近年，SAT技術を応用した高速 ASP

システムが開発され，スケジューリング，プランニン
グ，システム生物学，システム検証など様々な分野へ
の実用的応用が急速に拡大している．ハミルトン閉
路問題およびその関連問題に対して ASP を用いる利
点としては，ASP 言語の高い表現力，高速な解列挙，
組込み非閉路制約，充足不能コアに基づく最適化，イ
ンクリメンタル ASP解法などが挙げられる．
本論文では，解集合プログラミング (ASP)を用い



図 1 D. E. Knuth の教科書にある米国本土 48 州の隣接関係を表すグラフ

たハミルトン閉路問題，最短ハミルトン閉路問題，コ
スト制約付きハミルトン閉路問題の解法について述
べる．ハミルトン閉路問題を解く ASP 符号化とし
て，undirected，directed，acyclicity の 3つを考案し
た．undirected は，ハミルトン閉路問題を次数制約
と連結制約で簡潔に表現した符号化である．directed

は，与えられた無向グラフの各辺 u− v に対して，2

つの弧 u → v と v → u を対応させることで有向グ
ラフ化して解く符号化である．acyclicityは，directed

符号化をベースに，連結制約に代わる部分閉路禁止制
約を，ASP の組込み非閉路制約 [2]で表現した符号
化である．最短ハミルトン閉路問題とコスト制約付き
ハミルトン閉路問題については，考案した 3 つの符
号化に目的関数とコスト制約をそれぞれ追加するこ
とで自然に拡張できる．
考案した符号化の有効性を評価するために，Flinders

Hamiltonian Cycle Project (FHCP) で公開されてい
る HCPインスタンス (全 1001問) [5]を用いて実験
を行った．この問題集は FHCP Challenge と呼ばれ
る国際競技会で使用された頂点数が 66個から 9,528

個の HCP インスタンスから構成されている．実験の
結果，directed符号化が 875 問と最も多くの問題を解
き，他の符号化と比較してその優位性が確認できた．

2 ハミルトン閉路問題とその関連問題
グラフの全頂点を一度ずつ通る路はハミルトン路

(Hamiltonian Path)と呼ばれ，閉路はハミルトン閉
路 (Hamiltonian Cycle)と呼ばれる．以下に，本稿で
対象とする問題群について述べる．

• ハミルトン閉路問題は，与えられたグラフの全
頂点をちょうど一度ずつ通る閉路が存在するかど
うかを判定する問題である．

• 最短ハミルトン閉路問題は，グラフの辺に距離
が付随しているとき，最短距離のハミルトン閉路
を求める最適化問題である．

• コスト制約付きハミルトン閉路問題は，ハミル
トン閉路問題に，距離の総和が所与の閾値以下
(または以上) であることを制約条件として付加
した問題である [7]．

これらの問題から，始点と終点が一致するという閉路
の条件を取り除いた問題を，それぞれ，ハミルトン路
問題，最短ハミルトン路問題，コスト制約付きハミル
トン路問題と呼ぶ．
図 1 は，D. E .Knuth の The Art of Computer

Programming [10] に記載されている米国本土 48州
の隣接関係を表すグラフである．この隣接グラフの頂
点数は 48，辺の数は 105である．各辺に付与されて
いる数字は，州都の間の距離 (マイル)を表している．
西海岸ワシントン州 (WA)から東海岸メーン州 (ME)

までのハミルトン路は 6,876,928通りあり，最短ハミ
ルトン路は 11,698マイルである．コスト制約付きハ
ミルトン路問題は，WA から ME までの距離の総和
がある閾値以下であるハミルトン路を求める問題と
考えることができる．本稿では，コスト制約付きハ
ミルトン路問題の解を全列挙する問題を対象とする．
例えば，コスト制約を最短距離の 10%増 (12,868 マ
イル)以内とした場合，解の総数は 16,180 個である．



3 解集合プログラミング
解集合プログラミング (ASP; [1] [4] [8] [11]) の言語
は，一般拡張選言プログラムに基づいている．本稿で
は説明の簡略化のため，そのサブクラスである標準論
理プログラム (以降，論理プログラムと呼ぶ)につい
て説明する．以降，論理プログラムのソースコードは
すべて ASPシステム clingo †1 の言語構文に従う．
論理プログラムは，以下の形式のルールの有限集合
である．

a0 :- a1, . . . ,am,not am+1, . . . ,not an.

ここで，0 ≤ m ≤ n であり，各 ai はアトム，not

はデフォルトの否定 †2，カンマ (“,”)は連言を表す．
ピリオド (“.”)はルールの終わりを表す終端記号であ
る．“:-”の左側をヘッド，右側をボディと呼ぶ．ルー
ルの直観的な意味は，「a1, . . . , am がすべて成り立ち，
am+1, . . . , an のそれぞれが成り立たないならば，a0

が成り立つ」である．
ボディが空のルール (すなわち a0 :-.) をファク
トと呼び，“:-”を省略してもよい．また，以下のよ
うにヘッドが空のルールを一貫性制約と呼ぶ．

:- a1, . . . ,am,not am+1, . . . ,not an.

例えば，:- not a1,a2. は，「a2 が成り立つならば，
a1 が成り立つ」を意味する．
ASP言語には，組合せ問題を解くための便利な拡
張構文として，選択子と個数制約が用意されている．
例えば，選択子{a1; . . . ;an} をファクトとして書く
と，「アトム集合 {a1, . . . , an}の任意の部分集合が成
り立つ」を意味する．個数制約は選択子の両端に選
択可能な個数の上下限を付けたものである．例えば，
:- not lb {a1; . . . ;an} ub,b. と書くと，「bが成り立
つならば，a1, . . . , an のうち，lb個以上 ub個以下が
成り立つ」を意味する．
また，組合せ最適化問題を解くために，最小化関数

(#minimize) と最大化関数 (#maximize) 等も用意さ
れている．さらに，最近では，グラフ問題を解くため
の便利な拡張構文として，有向グラフの非閉路性を簡

†1 https://potassco.org/clingo/

†2 失敗による否定とも呼ばれる．述語論理の否定 (¬)
とは意味が異なる．

図 2 重み付き無向グラフの例

% node

node(1). node(2). node(3). node(4). node(5). node(6).

% edge

edge(1,2). edge(1,3). edge(2,4). edge(3,4).

edge(3,5). edge(4,6). edge(5,6).

% distance

cost(1,2,3). cost(1,3,2). cost(2,4,3). cost(3,4,2).

cost(3,5,1). cost(4,6,1). cost(5,6,2).

コード 1 グラフ (図 2) の ASP ファクト表現

潔に記述できる#edge宣言が追加されている [2]．例
えば，#edge(X,Y): arc(X,Y). は，arc(X,Y) を満
たす有向辺 X → Yを辺集合としてもつグラフが，閉
路をもたないことを保証する．

4 ハミルトン閉路問題のASP符号化
図 3に，ASP を用いたハミルトン閉路問題の解法
を示す．これは，与えられた問題インスタンスを ASP

ファクト形式に変換し，ハミルトン閉路問題を解く
論理プログラム (ASP 符号化) と結合した後，ASP

システムを用いて解を計算する一般的な解法である．
本節では，ハミルトン閉路問題のインスタンスを記
述するための ASP ファクト形式を説明した後，ハ
ミルトン閉路問題を解くための 3 つの ASP 符号化
undirected, directed, acyclicity について述べる．

4. 1 ASPファクト形式
最短ハミルトン閉路問題を例にとって，入力とな
る重み付き無向グラフ (図 2) の ASP ファクト形式
をコード 1 に示す．アトム node/1 は頂点，edge/2

は辺，cost/3は距離を表す．例えば，edge(1,2)は，
辺 1 – 2 を表し，cost(1,2,3) は，この辺の重みが



HCP 問題
インスタンス 変換器 ASP ファクト

ASP 符号化
(論理プログラム)

ASP システム HCP 問題
の解

- - - -

図 3 ASP を用いたハミルトン閉路問題 (HCP) の解法

1 { in(X,Y) } :- edge(X,Y).

2 :- not 2 { in(X,_) ; in(_,X) } 2, node(X).

3

4 reached(s).

5 reached(Y) :- reached(X), in(X,Y).

6 reached(Y) :- reached(X), in(Y,X).

7 :- not reached(X), node(X).

コード 2 undirected 符号化

1 { in(X,Y) ; in(Y,X) } 1 :- edge(X,Y).

2 :- not 1 { in(X,_) } 1, node(X).

3 :- not 1 { in(_,X) } 1, node(X).

4

5 reached(s).

6 reached(Y) :- reached(X), in(X,Y).

7 :- not reached(X), node(X).

8

9 :- not X < Y, in(s,X), in(Y,s).

コード 3 directed 符号化

3，すなわち，頂点 1 と頂点 2の距離が 3であること
を示している．

4. 2 undirected符号化
グラフ G = (V,E)にハミルトン閉路が存在する必
要十分条件は，以下の 2 つの制約を満たす部分グラ
フ G′ = (V,E′)が存在することである．
• G′ の各頂点の次数が 2 (次数制約)

• G′ が連結である (連結制約)

ハミルトン路の場合，次数制約は「始点と終点の次数
が 1，他の頂点の次数が 2」となる．
undirected 符号化は，ハミルトン閉路問題の次数
制約と連結制約を，ASP の一貫性制約と個数制約
を使って表した基本的な符号化である．コード 2 に
undirected符号化を示す．コード中の sは始点の頂点

番号を表し，実際の値は実行時に与えられる．
• 1 行目のルールは，各辺 edge(X,Y) に対して，
その辺がハミルトン閉路に含まれるかどうかを
意味するアトム in(X,Y)を選択子を使って導入
している．

• 次数制約は 2行目のルールで表される．このルー
ルは，各頂点 node(X)に対して，その次数が 2に
等しくなることを個数制約を使って表している．

• 連結制約は 4–7行目のルールで表される．ある頂
点 Xが始点 sから到達可能であることを意味する
補助アトム reached(X)を導入する．4行目のファ
クトは，始点 sが到達可能あることを表している．
5行目のルールは，各辺 X – Yに対して，その辺
がハミルトン閉路に含まれ (in(X,Y))，かつ，頂
点 Xが始点から到達可能であれば (reached(X))，
Yも到達可能であることを表している．6行目は
5行目と同様であるが，辺を逆向きに辿った場合
を表している．7行目のルールは，すべての頂点
が始点から到達可能でなければならないことを
一貫性制約を使って表している．

このように，undirected符号化は，ハミルトン閉路問
題の次数制約と連結制約を，ASPのルール 6個で簡
潔に表現している．特に，4–7行目のように連結制約
を簡潔に記述できる点は，ASP の大きな特長の一つ
である．

4. 3 directed符号化
directed 符号化は，undirected 符号化をベースに，
与えられた無向グラフの各辺 u− v に対して，2つの
弧 u → v と v → uを対応させることで有向グラフ化
して解く符号化である．コード 3に，directed符号化
を示す．
• 1行目では，無向グラフの有向グラフ化を行う．



1 { in(X,Y) ; in(Y,X) } 1 :- edge(X,Y).

2 :- not 1 { in(X,_) } 1, node(X).

3 :- not 1 { in(_,X) } 1, node(X).

4

5 #edge (X,Y): in(X,Y), X != s, Y != s.

6

7 :- not X < Y, in(s,X), in(Y,s).

コード 4 acyclicity 符号化

与えられた無向グラフの各辺 edge(X,Y) に対
して，in(X,Y)と in(Y,X)を導入し，二つの弧
X→Yと Y→Xのうち高々1個がハミルトン閉路に
含まれることを個数制約を使って表している．

• 次数制約は 2–3 行目のルールで表される．2行
目のルールは，各頂点 node(X)の出次数が 1に
等しいことを表している．同様に，3行目のルー
ルは入次数が 1に等しいことを表す．

• 連結制約は，directed 符号化と同様に，補助ア
トム reached/1を使って表される (5–7行目)．

• 有向グラフ化した場合，無向グラフ上の各ハミ
ルトン閉路に対して，右回りと左回りの 2 つの
対称なハミルトン閉路が存在する．9行目のルー
ルは，この対称解を除去する制約である．

4. 4 acyclicity符号化
acyclicity符号化は，directed符号化をベースに，連
結制約を部分閉路を禁止する制約に置き換えたもので
ある．コード 4に，acyclicity符号化を示す．directed

符号化との違いは，5行目だけである．
• 5行目のルールは，始点 sを含まない弧 in(X,Y)

の集合をもつ有向グラフが閉路を含まないこと
を，clingo の#edge 宣言を使って簡潔に表して
いる．

4. 5 最短ハミルトン閉路問題
最短ハミルトン閉路問題の ASP符号化は，これま
で説明した各符号化に，目的関数を追加することで実
現できる．undirected 符号化の場合は，コード 5 の
ルールを追加すればよい．この#minimize宣言は，ハ
ミルトン閉路に含まれる各辺 X – Yの距離 C の総和を
最小化することを意味する．directedと acyclicity 符

#minimize { C,X,Y: in(X,Y), cost(X,Y,C) }.

コード 5 目的関数

cost(Y,X,C) :- cost(X,Y,C).

コード 6 補助ルール

:- not #sum { C,X,Y: in(X,Y), cost(X,Y,C) } c.

コード 7 コスト制約

号化については，無向グラフの各辺 edge(X,Y)に対
して，2つの弧 in(X,Y)と in(Y,X)を導入している．
そのため，#minimize宣言に加えて，コード 6の補
助ルールを追加する必要がある．

4. 6 コスト制約付きハミルトン閉路問題
コスト制約付きハミルトン閉路問題は，ハミルトン
閉路問題に，距離の総和が所与の閾値以下であるこ
とを制約条件として付加した問題である．この問題
の ASP符号化は，ハミルトン閉路問題の undirected,

directed, acyclicity符号化に，コード 7のルールを追
加することで実現できる．コード中の c は閾値を表
し，実際の値は実行時に与えられる．また，最短ハ
ミルトン閉路問題の場合と同じ理由で，directed と
acyclicity 符号化については，アトム cost/3 に関す
る補助ルール (コード 6) も追加する必要がある．

5 実行実験
考案した 3 つの符号化の性能を評価するために実
行実験を行った．実験に使用したベンチマーク問題は
以下の通りである．
• ハミルトン閉路問題 (1,001問)

Flinders Hamiltonian Cycle Project (FHCP)で
公開されている HCPインスタンス (全 1,001問)

を使用した [5]．この問題集は FHCP Challenge

と呼ばれる国際競技会で使用された頂点数が 66

個から 9,528個の HCP インスタンスで構成され
ている．すべてのインスタンスはハミルトン閉路



図 4 ハミルトン閉路問題: カクタスプロット

が存在する SATインスタンスである．
• 最短ハミルトン路問題 (7問)

図 1に示した米国本土 48州の隣接グラフに基づ
く問題インスタンス (1問)と，2N + 1次の正方
グリッドグラフに基づく問題インスタンス (6問，
3 ≤ N ≤ 8)を使用した．グリッドグラフの辺の
重みはランダムに生成した．

• コスト制約付きハミルトン路問題 (全列挙)

図 1 に示した米国本土 48 州の隣接グラフに基
づく問題インスタンスを使用した．コード 7 に
示したコスト制約の閾値 c として，最短距離
(11,698マイル)の N 倍となる 10個の値，N =

1.00, 1.01, 1.02, 1.05, 1.1, 1.2, 1.3, 1.4, 1.5, 1.6 を
使用した．

ASP システムには，clingo のバージョン 5.5.0 を
使用し，実験環境は Mac OS (Intel Core i7 3.2GHz

64GBメモリ)である．

5. 1 ハミルトン閉路問題
表 1 に，ハミルトン閉路問題の解けた問題数を示

す．左から，HCPインスタンスの頂点数の範囲，問
題数，各符号化で解けた問題数を示している．各範
囲において，最もよい結果をボールド体で示してい
る．解けた問題数は，undirected 符号化が 715 問，
directed符号化が 875 問，acyclicity符号化が 473 問
であり，directed符号化がもっとも多くの問題を解い
た．directed符号化は，どの頂点数の範囲においても
他の符号化と同じかより多くの問題を解いており，そ

の優位性が確認できる．
図 4に，カクタスプロットを示す．縦軸は問題を解
くのに要した CPU 時間 (秒)，横軸は解けた問題数
を表す．グラフが右によるほど多くの問題を解けたこ
とを示し，下によるほどより速く解けたことを示す．
図 4より，directed符号化が，他の 2つの符号化と比
較して，より多くの問題を高速に解いていることが確
認できる．
今回の実験では，ASPシステム clingo のオプショ
ンは trendy を使用し，1問あたりの時間制限は 30分
とした．また，directedと acyclicity 符号化について
は，個数制約の上限 (コード 3と 4の 1行目) を省略
したものを使用した．

5. 2 最短ハミルトン路問題
表 2に，最短ハミルトン路問題について，各符号化

で得られた目的関数の値を示す．ASPシステム clingo

のオプションは trendy を使用し，1問あたりの時間
制限は 3時間とした．各問題ごとに，最も良かった値
をボールド体で示している．*マークは，最適値を表
している．最適値と最良値の数は，undirected符号化
が 5問，directed符号化が 3問，acyclicity符号化が 3

問であり，undirected符号化が 2問ではあるが，他の
符号化より精度の高い解を生成した．grid6 と US48

については，3つの符号化すべてで最適値を得ること
ができた．最適値を得るのに要した CPU 時間は，最
も速かった undirected 符号化で，grid6 は 1.451 秒，
US48 は 6.68秒であった．

5. 3 コスト制約付きハミルトン路問題
表 3 に，図 1 に示した米国本土 48 州の隣接グラ

フに基づくコスト制約付きハミルトン路問題につい
て，各符号化が解の全列挙に要した CPU 時間 (秒)

を示す．ASPシステム clingo のオプションは crafty

を使用し，1 問あたりの時間制限を 3 時間とした．
表の 1 列目はコスト制約の閾値と最短距離からの倍
率，2 列目は解の総数を表している．各閾値ごとに，
最も良かった値をボールド体で示している．表より，
acyclicity 符号化が，他の符号化と比較して，より多
くの問題を高速に解いていることがわかる．また，平



頂点数 問題数 undirected directed acyclicity

0 ≤ |V | < 1000 171 156 171 155

1000 ≤ |V | < 2000 165 120 158 124

2000 ≤ |V | < 3000 177 125 162 73

3000 ≤ |V | < 4000 185 104 148 42

4000 ≤ |V | < 5000 128 92 104 28

5000 ≤ |V | < 6000 80 63 68 23

6000 ≤ |V | < 7000 55 39 43 21

7000 ≤ |V | < 8000 28 12 14 5

8000 ≤ |V | < 9000 10 2 5 1

9000 ≤ |V | < 10000 2 2 2 1

合計 1001 715 875 473

表 1 ハミルトン閉路問題: 解けた問題数

問題名 頂点数 辺数 undirected directed acyclicity

grid6 49 84 *68,656 *68,656 *68,656

grid8 81 144 113,250 113,335 113,186

grid10 121 220 172,703 174,861 172,759

grid12 169 312 231,278 236,211 233,657

grid14 225 420 317,617 316,347 318,419

grid16 289 544 421,249 421,696 422,963

US48 48 105 *11,698 *11,698 *11,698

最適値または最良値の数 5 3 3

表 2 最短ハミルトン路問題: 得られた目的関数の値

閾値 (倍率) 解の総数 undirected directed acyclicity

11698(1.00) 1 2.979 7.531 4.586

11814(1.01) 8 5.587 15.322 5.250

11931(1.02) 28 3.243 18.600 3.578

12282(1.05) 388 10.003 19.818 6.296

12867(1.10) 16,180 16.548 28.555 9.764

14037(1.20) 939,209 48.262 40.717 26.837

15207(1.30) 4,525,541 88.172 55.276 42.037

16377(1.40) 6,702,964 99.154 47.647 40.640

17547(1.50) 6,876,526 95.390 45.265 38.411

18716(1.60) 6,876,928 98.937 49.138 40.748

平均 CPU 時間 46.8275 32.7869 21.8147

表 3 コスト制約付きハミルトン路問題: 解の全列挙に要した CPU 時間 (秒)

均 CPU時間も最も短い．

6 おわりに
本稿では，解集合プログラミング (ASP)を用いた
ハミルトン閉路問題，最短ハミルトン閉路問題，コス
ト制約付きハミルトン閉路問題の解法について述べ
た．ハミルトン閉路問題を解く ASP 符号化として，
undirected，directed，acyclicity の 3つを考案し，ハ
ミルトン閉路問題および関連問題の制約を，ASPの
ルール 10個程度で簡潔に表現できることを確認した．

考案した 3 つの符号化の有効性を評価するために，
Flinders Hamiltonian Cycle Project (FHCP) で公
開されているハミルトン閉路問題の問題集 (全 1001

問)を用いた実験を行った．その結果，directed符号
化が，875 問と最も多くの問題を解き，他の符号化と
比較してその優位性が確認できた．
本研究で使用したハミルトン閉路問題の問題集は，

2015 年から 2016 年にかけて行われた FHCP Chal-

lenge と呼ばれる国際競技会 [5]で使用されたもので
ある．この競技会では，整数計画ソルバー CPLEX



を使用した手法 [3] が 985 問を解き優勝した．2 位
となった SAT ソルバーを使用した手法は 614 問を
解いた [9]．3位から 5位の解けた問題数はそれぞれ，
488 問，464 問，385 問であった．よって，875 問を
解いた directed 符号化は，ASP のルール 10 個程度
の簡潔なプログラムであるにもかかわらず，実質 2位
の成績に相当する．
今後の課題としては，ハミルトン閉路問題の緩和
問題と CEGAR (Counterexample Guided Abstrac-

tion Refinement) を用いた解法 [13] [16] を応用した
提案手法の高速化，巡回セールスマン問題への拡張な
どが挙げられる．
ハミルトン閉路問題は代表的な NP 完全問題であ
り，また重要な工学的応用が数多く存在するため，そ
の解法については古くから盛んに研究されている．本
研究と関連の深いものとしては，SATソルバーを用
いた解法 [12] [14] [13]，整数計画ソルバーを用いた解
法 [15]がある．これらや FHCP Challenge の上位ソ
ルバー [3] [9] との比較も重要な今後の課題である．

